

The Maple command which follows defines a function piecewise. In the language of Laplace transforms,  $F(x) = \mathcal{U}(x - \frac{\pi}{3}) - \mathcal{U}(x - \frac{\pi}{2})$ . It is a block of height 1 in the interval  $[\frac{\pi}{3}, \frac{\pi}{2}]$  and is 0 otherwise.

```
>F:=x->piecewise(x<Pi/3,0,x<Pi/2,1,0);
```

I'll use this as initial data for the heat equation with temperature at the ends always 0. So we're looking for  $u(x,t)$  with  $u_t = u_{xx}$  and, for all  $t$ , both  $u(0,t) = 0$  and  $u(\pi,t) = 0$ .

**Interpretation** Would you believe a thin bar has temperature 0 except for a central chunk which has temperature equal to 1? This seems physically unlikely. This is more possibly an initial condition for diffusion, say a sugar solution in water. Adjust the units for concentration so that the highest concentration expected is 1 and a solution entirely water has concentration 0. Then we've considering a thin tube of water which has a high concentration of sugar in a central interval: maybe this is possible.

Separation of variables suggests that we compute the Fourier series of the function,  $F$ . We want a Fourier sine series which will be valid on the interval  $[0, \pi]$ . This Maple command gets the Fourier sine coefficients:

```
>g:=n->(2/Pi)*int(F(x)*sin(n*x),x=0..Pi);
```

Let's check:

```
>g(3);
```

$$-\frac{2}{3\pi}$$

The following instruction assembles a partial sum of the Fourier sine series for  $F$ :

```
>Q:=(N,t)->sum(g(n)*sin(n*x),n=1..N);
```


We can check it:

```
>Q(3);
```

$$\frac{\sin(x)}{\pi} + \frac{1}{2} \frac{\sin(2x)}{\pi} - \frac{2}{3} \frac{\sin(3x)}{\pi}$$

To the right is a picture of the 100<sup>th</sup> partial sum for  $F$  and  $F$  itself. The graph results from the Maple command

```
>plot({F(x),Q(100)},x=0..Pi);
```



Notice that Maple attempts to fit the graph into a square. The true aspect ratio (horizontal:vertical) is actually about 3-to-1.

Now let's look at graphs of solutions to the heat equation, approximating the initial data given by  $F$  on  $[0, \pi]$  with the zero boundary conditions. So we need a slight variation of the partial Fourier sum defined above. Here it is:

```
>QQ:=(N,t)->sum(g(n)*sin(n*x)*exp(-n^2*t),n=1..N);
```

And here is a test:

```
>QQ(3,t);
```

$$\frac{\sin(x) \exp(-t)}{\pi} + \frac{1}{2} \frac{\sin(2x) \exp(-4t)}{\pi} - \frac{2}{3} \frac{\sin(3x) \exp(-9t)}{\pi}$$


If this is too complicated, we can look at a fixed value of  $t$ :

```
>QQ(3,.01);
```


$$0.3151426498 \sin(x) + 0.1529143885 \sin(2.x) - 0.1939422210 \sin(3.x)$$

On the next page is the result of the following command for various values of  $t$ :


```
>plot({F(x),QQ(100,t)},x=0..Pi);
```




**t=.0001**




**t=.001**




**t=.01**



**t=.1**



**t=1**



**t=10**